



Races From Blocking “=” Assignments

Parallel Procedures

Blocking assignments follow order of statements inside a procedure

Parallel procedures have no order.

blocking

```
always @(posedge clk)
begin
  a = b;
end
```

```
always @(posedge clk)
begin
  b = a;
end
```

nonblocking

```
always @(posedge clk)
begin
  a <= b;
end
```

```
always @(posedge clk)
begin
  b <= a;
end
```

Parallel procedures

blocking


- With Blocking both start at same time.
- a could transfer to b first.
b could transfer to a first.

nonblocking

- This is two parallel flip-flops
- Both clocked at same time.

Think: next-state \leftarrow previous state

$$a^+ \leftarrow b;$$
$$b^+ \leftarrow a;$$



Races From Using Blocking Assignments

Inside a procedure **blocking** assignments are executed in the order of the statements

```
always @(b)
  begin
    a=b;
    c=a;
  end
```

Makes c=b;

Inside a procedure **nonblocking** assignments always use the data valid before the trigger.

```
always @(b)
  begin
    a<=b;
    c<=a;
  end
```

is equivalent to c<=a; a<=b;

Parallel procedures in different always blocks, have no predefined order with **blocking** assignments.

```
always @(b)
  a=b;
always @(b)
  c=a;
```

May make c=b; or it may make c=a; //for some old a

Parallel procedures in different always blocks are defined with **nonblocking** assignments.

```
always @(b)
  a<=b;
always @(b)
  c<=a;
```

Will make c=a; //for some old a



Multiple Assignments

Two Outputs Connected Together

```
always @(posedge Clk)
begin
  if (En1) Q<=D1;
end
```

```
always @(posedge Clk)
begin
  if (En2) Q<=D2;
end
```

Multiple Assignment

- **Mutually exclusive** means that $Q \leq D1$ and $Q \leq D2$ could never happen together.
- If they were both in one if-else statement the compiler would know they could never happen together.
- Here both EN1 and EN2 might be true together.

Possible Simulation Results

- The simulator will choose one to do first. No one knows which. The lasting result will be the final one.

Possible Synthesis Results

- The compiler chooses one result.
- The compiler generates two flip-flops and ANDs the result.
- All outputs may be disconnected.

Multiple Assignments


If both statements are in the same procedure, the En2 would replace the En1 result in zero time. In synthesis this would mean the En2 result would take priority over En1.

```
always @(posedge Clk)
begin
    if (En1) Q=D1;
    if (En2) Q=D2;
end
```

Blocking was used for flip flops, to ensure Q=D2 is done after Q=D1 and hence replaces Q=D1.

If delays are put on the statements simulation could give a glitch. Synthesis would not. It would generate a circuit which would give EN2 priority.

```
always @(posedge Clk)
begin
    if (En1) #2 Q<=D1;
    if (En2) #3 Q<=D2;
end
```



20. • PROBLEM
What happens here?¹

```
always @(posedge Clk)
begin
    if (En1) Q=D1;
end

always @(posedge Clk)
begin
    if (~En2) Q=D2;
end
```

1. This is a poor way to code. It should work for simulation since the two Qs are never activated at once. Synthesis might do anything.



Using Case Statements

Demux Inference from case

```
wire [2,0] in;  
reg [7:0] Y;  
  
always @(in) begin  
  case(in)  
    3'd0: Y=8'b00000001;  
    3'd1: Y=8'b00000010;  
    3'd2: Y=8'b00000100;  
    3'd3: Y=8'b00001000;  
    3'd4: Y=8'b00010000;  
    3'd5: Y=8'b00100000;  
    3'd6: Y=8'b01000000;  
    3'd7: Y=8'b10000000;  
  endcase
```

I would put in a default statement, even though it is not needed.
default: \$display (" Error");

Full Case


- All cases are covered.
It avoids latches

Parallel Case (Mutually Exclusive)

- No two cases can be active at once.
It can be implemented as a mux.

No undefined or parallel cases

- The compiler can statically determine that all possible cases are covered.
- The compiler can statically determine that no two cases are active at once.



Multiple Assignment Race (cont. from previous page)

21. • PROBLEM

What common simulation problem might be caused by this code?

```

wire clk, x, y;
reg m,q;
always @(posedge clk)
  begin: storage
    q <= m;
  end

```


```

always @(clk or x or y)
  begin: mux
    if (clk) m = x;
    else m = y;
  end;

```

BAD

Here the mux needs nonblocking¹!



1. Another way of insuring the mux is done last is to write:

`if (clk) #0 m=x; else #0 m=y;` “#0” declares that a statement is to be done last in that time step. See S. Palitkar, *Verilog HDL*, p. 127.



Full Case; Latch Inference in Case

Not Obvious Full-Case,
Case with a restricted input

```
reg [6:1] Y;  
  
always @(a or b or c)  
begin  
  // if a,b,c = 1,1,1 make c1=0  
  c1 = (a&b&c) ? 0 : c  
  case ({a,b,c1})  
    3'd0: Y=000000;  
    3'd1: Y=000001;  
    3'd2: Y=000010;  
    3'd3: Y=000100;  
    3'd4: Y=001000;  
    3'd5: Y=010000;  
    3'd6: Y=100000;  
  endcase  
end
```

Apparent undefined cases

3'd7: Y=000000;

a,b,c = 1,1,1 cannot occur.
Synthesis does not know that.
Thus synthesis will infer 7 latches.

To avoid latches

Put in default

```
....  
3'd5: Y=010000;  
3'd6: Y=100000;  
default: Y=000000;
```

Better default

A better default is:

default: Y=xxxxxx;

It gives the synthesizer more choice.

Using Case

There are several problems that can happen with case statement synthesis.

1. If the case is known to cover all the possibilities the input condition can assume it is said to be a *full case*. Unfortunately the synthesizer will not know this unless case covers all 2^N possibilities for an N bit condition. If the synthesizer does not know it is a full case, it will insert latches.
2. If two different conditions may happen at once, they will activate two different outputs at the same time. This is called a *nonparallel case*.

Not Obvious Full-Case

If a *case* contains all 2^n cases no latches will be generated.

If it contains less than 2^n cases, latches will be generated unless it is very obvious all cases are covered. Synthesizers do not look back very far to determine if all cases are covered.

Use Default

The default statement does no harm if it is put in and not needed.

Always put in a default, whether you need it or not, unless you want the latches.

Place xxxx as a the Default Output

If you know the default will never be selected by the case, then you can put in anything you want. The logic that is easiest to minimize is x (don't care). Requiring the default to take some particular value, like zero, can greatly increase the size of the circuit.



Nonparallel Case

NOT Mutually Exclusive

```

always @(x or y or z) begin
  case (1'b1)
    x:      Y=2'b01;
    y:      Y=2'b10;
    z:      Y=2'b11;
    default: Y=2'bxx;
  endcase

```

FORCE Mutually Exclusive (Parallel)


```

always @(x or y or z) begin
  case ({x,y,z})
    3'b100:      Y=2'b01;
    3'b010:      Y=2'b10;
    3'b001:      Y=2'b11;
    default:     Y=2'bxx;
  endcase

```

Not Mutually Exclusive (nonparallel) Case

- Two cases can be active at once. Priority encoder generated.



Synopsys Compiler Directive

Simulator treats as comments
Tells Synopsys x, y and z can never happen at the same time.

Designer must enforce this!

Force Parallel Case

```

// synopsys parallel_case
avoids creating a priority encoder

case (1'b1) // synopsys parallel_case
  x:      Y=2'b01;
  y:      Y=2'b10;
  z:      Y=2'b11;
  default: Y=2'bxx;
endcase

```

Not Recommended


Parallel-Case

Whenever two or more lines of the case statement may be selected at once, the simulation executes the first line encountered in the listing. This is like a priority encoder.

In a *parallel* case, the synthesizer assumes some other circuit keeps two lines from being selected at once.


//synopsis directives can be used to tell the synthesizer to force a parallel-case but one can also write the code to explicitly say what is desired. This latter method is synthesizer independent and keeps the simulation and synthesis in agreement. Such code may require the *casex* (or *casez*) command described on the next slide.

Coding for full decoding, priority encoder, or parallel case




```
always @(x or y or z)
begin
case({x,y,z})
3'b100 : Y=3'b100;
3'b010 : Y=3'b010;
3'b001 : Y=3'b001;
default: Y=3'b000;
endcase
```

Full decoding:
Assumes more than one of x, y, z can be 1 and removes those cases.



```
always @(x or y or z)
begin
casex({x,y,z})
3'b1xx : Y=3'b100;
3'b01x : Y=3'b010;
3'b001 : Y=3'b001;
default: Y=3'b000;
endcase
```

Priority decoder
Assumes more than one of x, y, z can be 1 but takes the first one as correct



```
always @(x or y or z)
begin
casex({x,y,z})
3'b1xx : Y=3'b1xx;
3'bx1x : Y=3'bx1x;
3'bxx1 : Y=3'bxx1;
default: Y=3'bxxx;
endcase
```

Parallel case
Assumes only one of x, y, z can be 1 at a time. Depends on some other circuit to enforce this.



Difference between case, casez and casex

Case treats 1,0,x,z as separate; x matches only x; x matches only z.

Casez treats z as a wild card. It matches 1, 0, x and z.

Casex treats both x and z as wild cards; either will match anything

```
case( )
1: matches 1
0: matches 0
z: matches z
x: matches x
```

1x0

↓
case()
110: no
x10: no
1z0: no
1x0: match

*x, z on left
are not wild cards*

```
casez( )
1: matches 1
0: matches 0
x: matches x
z: matches z,x,0,1
```

1x0

↓
casez()
110: no
x10: no
1z0: match
1x0: match

*z on left
matches anything*

```
casex( )
1: matches 1
0: matches 0
x: matches z,x,0,1
z: matches z,x,0,1
```

1x0

↓
casex()
110: match
x10: match
1z0: match
1x0: match

Casex/casez

For simulation

Case treats a bit in a variable as having four possible values {0, 1, x, z}, thus x only matches x, not 1 or 0.

Casex treats x, z or ? as a don't care which can match 0, 1, x or z.

Casez treats z or ? as a don't care which can match 0, 1, x or z, but x cannot match 0 or 1.

		case			
case\data		0	1	x	z
0		1	0	0	0
1		0	1	0	0
x		0	0	1	0
z,?		0	0	0	1

		casex			
casx\data		0	1	x	z
0		1	0	1	1
1		0	1	1	1
x		1	1	1	1
z,?		1	1	1	1

		casez			
casez\data		0	1	x	z
0		1	0	0	1
1		0	1	0	1
x		0	0	1	1
z,?		1	1	1	1

For example: Given- aa = 3b'1x0;

case (aa)

3'b110: ...// No match because 1 does not match x with a *case* statement.

3'b1x0: ... // Matches

casex (aa)

3'b110: ... // Matches aa because 1 does match x with a *casex* statement.

3'bx10: ...// Matches aa.

casez (aa)

3'bxx0: ... // x does not match 1 with a *casez* statement, although x matches x.

3'bzz0: ...// Matches aa.

For synthesis

No x values ever propagate in synthesis. However x values in the simulation cause an unexpected match with *casex*. Using *casez* will avoid those problems.

Don't cares in the outputs are fine for *case*, *casez* or *casex*.



Don't Cares In Case Control-Items

```
wire [3,1] in;
reg [1:0] Y;
```

```
always (in) begin
  casez(in)
```

// With casez, in=11x will match only default.

```
  3'bzz1: Y=2'b01;
  3'bz10: Y=2'b10;
  3'b100: Y=2'b11;
  default Y=2'b00;
endcase
```

Don't Cares In Right-Hand Side


```
always @(x or y or z) begin
  case ({x,y,z})
    3'b001: Y=2'b01;
    3'b010: Y=2'b10;
    3'b100: Y=2'b11;
    default Y=2'bxx;
  endcase
```

Generates a priority encoder

- zz1 has the highest priority.

Default

- Covers only 3'b000.



Don't Cares Can Simplify Logic

- Don't force the defaults to zero if you don't care. It makes the logic larger.
- Casez not necessary for output don't cares.
- casez preferred over casex.

Casez will stop unknowns in simulation

With `casez`, `x` inputs will give an error message.

```
wire [2,0] ct;  
reg [1:0] Y;
```

```
always (ct) begin: Proc-Safey  
    casez(ct)
```

```
// With casez, ct = 1x will match only default.
```

```
    3'bz1: Y=2'b01;
```

```
    3'b10: Y=2'b10;
```

```
    3'b00: Y=2'b00;
```

```
    default $display("11 or x input in Proc-Safey")
```

```
endcase
```

```
always (ct) begin: Proc-Sloppy
```

```
    casex(ct)
```

```
// With casex, ct = 1x will match 3'bx0 (also 3'b10).
```

```
    3'bx0: Y=2'b01;
```

```
    3'b10: Y=2'b10;
```

```
    3'b00: Y=2'b00;
```

```
    default $display("11 input in Proc-Sloppy")
```

```
endcase
```



Negative Numbers

Confusion Between Reg and Integer

Integers are 2's Complement

Integers declarations default to a 32-bit 2's complement number.
The compiler will eventually decide how many bits are needed.

Reg numbers are nonnegative integers

The length of registered numbers is given in the declaration.

Negative register numbers sign extend only to the length of the register

+3	011
+2	010
+1	001
0	000
-1	111
-2	110
-3	101
-4	100
3-bit 2's complement numbers	

```

reg [7:0] B,C;
reg [4:0] X;
always @(B, C, X)
begin
  X = - 5'd7;
  B = 10;
  C = B + X;
end

```

Value of X
X will hold a 5-bit -7
-00111 ⇒ 11000 +1 ⇒ 11001 {-7 in 2's complement}
Reg numbers are never negative,
Hence 11001 is taken as 25.

B will hold 00001010 (10 truncated to reg[7:0])

B = 0000,1010	10
+ X = + 1,1001	25
C = 0010,0011	35

Did you want 35 or 3?

Negative Numbers

Two's Complement

To change a binary number to its two's complement

Change the exchange the ones and zeros, then add 1, ignore any off-end carries from the add.

$-10 \Rightarrow -001010 \Rightarrow 110101 + 1 \Rightarrow 110110 \{-10 \text{ in 2's complement}\}$

Sign Extension

Two's complement numbers of the same length may be added in a normal adder.

Two's compliment numbers of different lengths must be sign extended when added.

The leftmost bit is replicated until the words are the same length. For example

$$\begin{array}{r}
 0000,1010 \quad \text{sign extend} \quad 0000,1010 \quad (10) \\
 + \quad 1,1001 \quad \Rightarrow \quad + \quad 1111,1001 \quad (-5) \\
 \hline
 \cancel{0010,0011} \quad \quad \quad 0000,0011 \quad (3)
 \end{array}$$

A one-bit sign extension in Verilog might be written:

```
reg [4:0] x;   reg [5:0] y, z;
      z = {x[4], x} + y ; // sign extend x to 6 bits.
```

If the bits represent unsigned numbers, then do not sign extend.

Sign Extension and Overflow

Two's complement addition is subject to overflow.

If one sign extends before adding, one can never get overflow.


$$\begin{array}{r}
 010 \quad (2) \quad \text{sign extend} \quad 0 \ 010 \quad (2) \\
 + 011 \quad (3) \quad \Rightarrow \quad + \quad 0 \ 011 \quad (3) \\
 \hline
 \cancel{101} \quad (-3) \quad \quad \quad 0101 \quad (5)
 \end{array}$$



Using For Loops For Building Iterative Hardware

Build an W-bit ripple-carry adder.

```
parameter W = 5; // Take W=5 here
reg [W-1:0] A, B, reg [W:0] S; // S is the sum
wire c_in, c; // c changes its meaning in each loop
integer i; // always make the index an integer
always @(c_in or A or B)
begin
  c = c_in;
  for(i=0; i < W; i=i+1)
  begin
    {c, S[i]} = A[i] + B[i] + c // Concatenate the outputs into a 2-bit vector.
  end
  S = {c: S(W-1:0)};
end
```



Hardware Loops¹

Loops give multiple copies of a basic instance.

The code in the loop will be synthesized, a different instance for each iteration.

Output leads from one block, with the same name as an input lead, will connect between iterations.

See the variable “c” in the program.


The index variable “i” is not changed outside the loop, so it should not be in the trigger list.

While loops are partially supported for synthesis. They represent a conditional branch. All while loops must be broken by an @(posedge clock) statement. Thus:-

```

always @(posedge clock)
  begin
    while ( b < 8 )
      begin
        @(posedge clock) ; // break the zero delay loop
        b <= b+2;
      end
    end
  end

```



1. Palnitakar, *Verilog*, Prentice Hall, 1998, p..285




An Iterative Comparator Hardware

Build an 5-bit comparator from blocks.

```

reg [4:0] A, B;
reg x, y;
integer i;
always @(A or B)
begin
  x=0; y=0; // Above the highest order bit, the two are equal
  for(i=4; i>=0; i=i-1) // x=0, y=0 goes in at high end.
  begin
    x=(A[i] > B[i])&(~y) | x; // A is larger at this bit or at a higher order bit.
    y=(B[i] > A[i])&(~x) | y; // B is larger at this bit or at a higher order bit.
  end
end
end

```



At the output:

$x,y = 0,0$ means $A=B$, $x,y = 1, 0$ means $A > B$, $x,y = 0,1$ means $A < B$,

Loops to Generate Iterative Circuits

This is an iterative comparator used as a lab in the Switching Circuits course at Carleton.

It only compares non-negative integers, where the number with the leftmost “1” is the largest.

22. • PROBLEMS

a. Write a **for** loop to calculate the parity of a 6-bit number. It should include-

```
if (data[i]) OddPar= ~OddPar;
```

b. One way to change a binary number to its two’s complement is:

Start at the right hand side.

Leave all bits unchanged until after the first “1” is found.

Invert all bits to the left of the initial “1”.

Thus: 1001_1000 has complement 0110_1000

Write a loop to generate such a circuit.



Compiler Directives

Tell The Synthesizer What To Do

- Written like comments
`// synopsys . . .`
- The simulator will ignore them
- Directs synthesis.
- Simplifies some language problems.
However it is nearly always possible to avoid them by proper coding.
- Thus simulation will agree with synthesis only if it was coded properly.
- Limits you to one compiler.
- Makes formal verification difficult.
- There are many of these compiler directives.
Check the Synopsys Manual

Example

Force Asynchronous Reset

```
module latch(Q,D,C,R);  
  input D,C,R;  
  output Q; reg Q;  
  
  //synopsys asynch_set_reset "R"  
  always @(C or R)  
  begin:  
    if (R)  
      Q = 0;  
    else if (C)  
      Q = D;  
  end  
endmodule
```

Compiler Directives

Other Compiler Directives

// synopsys async_set_reset

This one is needed to synthesize proper asynchronous resets.

// synopsys sync_set_reset

//synopsys async_set_reset_local

applies directive to specified signals in a named block

//synopsys one_hot

indicates only one of a list of signals is true at a time.

Useful to show set and rest are never both applied at once.

One of the more useful compiler directives is used to force a particular library module for arithmetic operations (next slide).

Formal verification

This is where the logic of a program is compared weith the logic of another program. This is often done after inserting special structures only used for testing, or after had optimizations on a compiled circuit.

The verification programs have trouble with compiler assertions.



Forcing Specific Synopsys Designware

Synopsys uses designware to implement counters, adders, comparators, etc.
Control the type of function used by inserting compiler directives into your code.

Example:

Library DW01 has two increments, ripple carry “rpl” and carry look-ahead “cla.”
Force the named block `bill` to use a carry look-ahead incrementer.

```
always @(price)
begin : bill //named procedure
    /* synopsys resource billspecial:
       map_to_module = "DW01_inc",
       implementation = "cla",
       ops = "greasedIncr";
    */
    newprice = price + 1; //synopsys label greasedIncr
end
```

- Must insert only in a nonclocked, named procedure or function.
i.e not after `@(posedge. . .`,
- `billspecial` will be the name given this instantiation.
- `"DW01_inc"` and `"cla"` are from the Synopsys library DW01
- The label applies to the most recently parsed function.
`newprice = price + 1 // synopsys label greasedIncr`

Mapping to a Specific Library Module

You may not need this

Both Synopsis and Ambit (PKS) will select simple circuits like adders, to meet your constraints.

Named Procedures

Specifically map an operation it must be inside a named procedure. named by writing the name after begin.

```
always @(a or ...  
    begin: bill  
    ...
```

Meanings of the mapping labels

// synopsis label greasedIncr labels the + operation with name *greasedIncr*.

This label is bound to the instantiation named *billspecial* by the *ops = "greasedIncr"*; statement.

The resource is module *DW01_inc*, in the designware library *DW01*

The specific implementation in the library is *cla*.

Libraries are fairly automatic

The simulator will automatically choose an implementation for your criteria.

Experience with Library Adders

The DW01 library has (1999) had five adders. For a 4 to 7 bit adds in a Viterbi decoder, a Carleton graduate student, Youxing Zhao found:

The conditional sum adder (*csa*) was the fastest.

The ripple carry adder (*rpl*) was second and significantly slower.

The fast carry look-ahead (*clf*) was third.

The Brent-Kung (*bk*) and the carry look-ahead adder (*cla*) were last and about the same.



Summary

Guidelies:

- Partition FSMs into next-state calc, outputs and registers.
- Use `<=` in the register procedure; use `=` in the others.
- In procedures:
 - Feed all right-hand side variables through the trigger list (unless also on the left side.)
 - Make all branches evaluate all left-hand side variables.
- If you are using negative numbers, add/sub only registers of equal length, and do sign extensions.
- Do not have the same left-hand side variable stored in two different procedures.
- For case statements:
 - Always use a *default* at the end. The default can be `$display("error")`
 - Use `casez` if there are don't cares in the control.
 - Use `x` for don't care outputs to minimize logic.
- If you want a priority encoder, use `casez` and conditions like `xxx1, xx10, x100, . . .`
- Flip-flops procedures must start `@(...edge clk)` or `@(...edge clk or ...edge reset)`
- Never reset combinational logic.
- A "rst" signal that is initially 1, may not be seen as having a rising edge at zero.
- Do not use *initial*, except in your test bench.

Mapping to a Specific Library Module
